Code 81X82-24983: 

“Apply a Route Layer's Definition Query to a Route Event layer”-ArcGIS 8.1.x & 8.2 only
Public Sub ApplyRouteLayerDefQueryToRouteEventLayer()

  On Error GoTo eh

  ' works by setting the Route Event's IRouteLocator::RouteWhereClause to be the same as

  ' the Route layer's Definition Query (IFeatureLayerDefinition::DefinitionExpression).

  '

  ' start by checking the user has selected both required layers in the TOC (6 tests).

  ' test 1: if nothing is selected in the TOC.

  Dim pMxDoc As IMxDocument

  Set pMxDoc = ThisDocument

  If pMxDoc.SelectedItem Is Nothing Then

    MsgBox "You must select the Route layer and the Route Event layer in the Table of Contents first.", _

            vbCritical, "No layers have been selected in the Table of Contents"

    Exit Sub

  End If

  ' test 2: if only 1 item is selected in the TOC.

  Dim pUnk As IUnknown

  Set pUnk = pMxDoc.SelectedItem

  If Not TypeOf pUnk Is ISet Then

    MsgBox "You must select both a Route layer and a Route Event layer in the Table of Contents.", _

            vbCritical, "Only one item has been selected in the Table of Contents"

    Exit Sub

  End If

  ' test 3: if more than 2 items are selected in the TOC.

  Dim pTheSet As ISet

  Set pTheSet = pUnk

  If pTheSet.Count <> 2 Then

    MsgBox "You must select only a Route layer and a Route Event layer in the Table of Contents.", _

            vbCritical, "Too many items selected in the Table of Contents"

    Exit Sub

  End If

  Dim pFirstSelItem As IUnknown

  Dim pSecondSelItem As IUnknown

  pTheSet.Reset

  Set pFirstSelItem = pTheSet.Next

  Set pSecondSelItem = pTheSet.Next

  ' test 4: make sure both selected items are feature layers.

  If Not TypeOf pFirstSelItem Is IFeatureLayer Or Not TypeOf pSecondSelItem Is IFeatureLayer Then

    MsgBox "You must select both a Route layer and a Route Event layer in the Table of Contents.", _

            vbCritical, "One or both of the selected items in the Table of Contents is not a Feature Layers"

    Exit Sub

  End If

  ' test 5: one of the selected items must be a route event layer.

  Dim blnFirstSelItemIsRouteEvent As Boolean

  If IsRouteEventLayer(pFirstSelItem) = True Then

    blnFirstSelItemIsRouteEvent = True

  End If

  If blnFirstSelItemIsRouteEvent = False Then

    If IsRouteEventLayer(pSecondSelItem) = False Then

      MsgBox "Neither of the selected items in the Table of Contents is a Route Event layer.", _

            vbCritical, "No Route Event layer"

      Exit Sub

    End If

  End If

  ' test 6: the other selected item must be a route layer.

  If blnFirstSelItemIsRouteEvent = True Then

    If IsRouteLayer(pSecondSelItem) = False Then

      MsgBox "Neither of the selected items in the Table of Contents is a Route layer.", _

            vbCritical, "No Route layer"

      Exit Sub

    End If

  Else

    If IsRouteLayer(pFirstSelItem) = False Then

      MsgBox "Neither of the selected items in the Table of Contents is a Route layer.", _

            vbCritical, "No Route layer"

      Exit Sub

    End If

  End If

' get each required layer (the user can select the 2 layers in any order).

  Dim pRouteLayer As IFeatureLayer

  Dim pRouteEventLayer As IFeatureLayer

  If blnFirstSelItemIsRouteEvent Then

    Set pRouteLayer = pSecondSelItem

    Set pRouteEventLayer = pFirstSelItem

  Else

    Set pRouteLayer = pFirstSelItem

    Set pRouteEventLayer = pSecondSelItem

  End If

  ' we will soon need to know if the Route Event has an existing RouteWhereClause. this

  ' allows it to be cleared if the Route layer's def query was just cleared.

  Dim pRouteEventSource As IRouteEventSource

  Set pRouteEventSource = pRouteEventLayer.FeatureClass

  Dim pRouteLocator As IRouteLocator

  Set pRouteLocator = pRouteEventSource.RouteLocator

  Dim strExistingRouteWhereClause As String

  strExistingRouteWhereClause = pRouteLocator.RouteWhereClause

  ' get the route layer's def query.

  Dim pDefQuery As IFeatureLayerDefinition

  Set pDefQuery = pRouteLayer

  Dim strExistingDefinitionExpression As String

  strExistingDefinitionExpression = pDefQuery.DefinitionExpression

  ' raise error if the donor Route layer does not have a def query, unless the Route Event

  ' has a RouteWhereClause, in which case it should be cleared.

  Dim blnClearExistingRouteWhereClause As Boolean

  If strExistingDefinitionExpression = "" Then

    If strExistingRouteWhereClause <> "" Then

      blnClearExistingRouteWhereClause = True

    Else

      MsgBox "The Route layer '" & pRouteLayer.Name & "' does not have a Definition Query.", _

             vbExclamation, "Can not proceed"

      Exit Sub

    End If

  End If

  ' copy it into the route event source's RouteLocator object.

  If blnClearExistingRouteWhereClause = False Then ' if Route layer has new query def.

    pRouteLocator.RouteWhereClause = strExistingDefinitionExpression

  Else

    pRouteLocator.RouteWhereClause = "" ' if Route layer's query def was just cleared.

  End If

  pMxDoc.ActiveView.Refresh

  Exit Sub

eh:

  MsgBox Err.Number & ": " & Err.Description, vbCritical, "Can not proceed"

End Sub

Private Function IsRouteEventLayer(pUnk As IUnknown) As Boolean

  ' returns True if the selected layer is a Route Event layer.

  IsRouteEventLayer = False

  Dim pFeatLayer As IFeatureLayer

  Set pFeatLayer = pUnk

  Dim pFeatClass As IFeatureClass

  Set pFeatClass = pFeatLayer.FeatureClass

  If Not TypeOf pFeatClass Is IEventSource Then

    Exit Function

  End If

  IsRouteEventLayer = True

End Function

Private Function IsRouteLayer(pUnk As IUnknown) As Boolean

  ' returns True if the selected layer is a Route layer. we must add IEventSource

  ' test to exclude existing line Route Event layers, since they are also Polylines with M's.

  IsRouteLayer = False

  Dim pFeatLayer As IFeatureLayer

  Set pFeatLayer = pUnk

  Dim pFeatClass As IFeatureClass

  Set pFeatClass = pFeatLayer.FeatureClass

  If TypeOf pFeatClass Is IEventSource Then

    Exit Function

  End If

  Dim strShapeFieldName As String

  strShapeFieldName = pFeatClass.ShapeFieldName

  Dim pField As esriCore.IField

  Set pField = pFeatClass.Fields.Field(pFeatClass.FindField(strShapeFieldName))

  Dim pGeomDef As IGeometryDef

  Set pGeomDef = pField.GeometryDef

  If pGeomDef.GeometryType <> esriGeometryPolyline Then

    Exit Function

  End If

  If pGeomDef.HasM = False Then

    Exit Function

  End If

  IsRouteLayer = True

End Function

